unittest
{
	assert( [ "5 3", "-10 10 -10 10 -10" ].parse.expand.solve == 10 );
	assert( [ "4 2", "10 -10 -10 10" ].parse.expand.solve == 20 );
	assert( [ "1 1", "-10" ].parse.expand.solve == 0 );
	assert( [ "10 5", "5 -4 -5 -8 -4 7 2 -4 0 7" ].parse.expand.solve == 17 );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nk = input.front.split.to!( long[] );
	input.popFront;
	auto as = input.front.split.to!( long[] );
	return tuple( nk[ 1 ], as );
}

auto solve( long k, long[] as )
{
	// 全体を白or黒で塗る⇒範囲を狭めて逆の色で塗る⇒更に範囲を狭めて逆の色で塗る⇒(繰り返し)
	// のように塗ると、最後のkマス以外は連続するマス数を気にせずに好きな色で塗れる
	
	// 最後のkマスを全検索してスコアの最大値を計算
	auto ans = 0L;
	auto t  = as[ 0 .. k ].sum;
	auto lp = 0L;
	auto rp = as[ k .. $ ].filter!( ( a ) => 0 < a ).sum;
	for( auto li = 0L, ri = k;; li++, ri++ )
	{
		ans = max( ans, lp + rp + ( ( 0 < t ) ? t : 0 ) );
		if( ri == as.length ) break;
		t += as[ ri ] - as[ li ];
		if( 0 < as[ li ] ) lp += as[ li ];
		if( 0 < as[ ri ] ) rp -= as[ ri ];
	}
	return ans;
}
